二、控件开发纵览
   通过开发上边这个控件，我们已经对Delphi控件开发有了基本的认识。下面我们将系统的讲述一下控件开发的知识。

制作控件第一件事就是选择适当的 Delphi对象类型作为父对象，以派生新的对象。子对象可以继承父对象的全部非 private部件，但不能摆脱不需要的部件。因此，所选父对象应尽可能多地包含子对象所需的属性、事件和方法，但不应包含子对象不需要的东西。Delphi必须从Tcomponent或Tcomponent的子类派生。TComponent 是所有 Delphi控件的基点，但若直接从 TComponent 派生新控件，很多东西就需要自己从头做起。一般只有非可视控件才直接从 TComponent派生。 Delphi提供了若干专门用于制作控件（可视控件）的对象类型，都是从 TControl和 TWinControl派生而来。

TControl的子类型用于非窗口式控件， TWinControl的子类型则用于窗口式控件。除非特殊需要，一般不直接从 TControl和TWinControl派生新控件，而是从其子类型派生。这样可以充分利用原有的属性、事件和方法，减少很多工作量。在这些控件类型 中，非通用的属性、事件和方法都声明为 protected。这样可以禁止控件用户访问，又能被子类型继承和修改。在新控件中，可以简单地把继承来的属性和事件重新声明为 published，使控件用户能在设计期通过对象编辑窗口访问，也可以进而修改属性的默认值和读写方式，或是重载（ override）事件处理子过程和其他控件方法，以修改其中的程序代码。重声明可以放宽访问权限，但不能相反，例如，不可能把 published属性重声明为 private或 protected。
Delphi控件也是Delphi的类，所有的控件都有特定的结构。一般控件包括三大组成部分：属性、方法和事件，下面先介绍初学控件开发的最难懂的属性部分，其他部分我们将在以后章节为大家介绍。

属性主要部分就是属性的读写方法（或读写字段）。前面的例子用的是读写字段，也就是对属性的读写都通过对字段的读写来完成。下面为大家讲解一下读写方法的使用方法：

TmyComponent = class(TComponent)

Private

  Fcount: Integer;

  Procedure SetCount(Avalue: Integer);

Pulbished

  Property Count: Integer read Fcount write SetCount;

End;

这个例子中当执行MyComponent1.Count := 1;这样的代码时，将会导致SetCount方法执行，并且参数Avalue被指定为1；当执行 I := MyComponent1.Count;方法时，会将 Fcount的值返回给I。

属性的声明语法允许属性声明的Read和Write部分用访问方法取代对象私有数据域。属性的读方法是不带参数的函数，返回同属性相同类型的值。通常读方法以Get开头。属性的写方法总是带一个参数的过程。写方法常常以Set开头。
三、开关控件TlincoSwitch
用过Delphi1（好古老的东东呀！）的人相信都记得这个开关控件 
1、建立位图资源文件：

用Image Editor建立一个Res文件，并在文件中分别建立下面两个位图[image: image1.jpg]

，并分别命名为SWITCHON、SWITCHOFF。保存此Res到控件单元所在目录下。

2、写控件代码。

unit LincoSwitch;

interface

uses

  SysUtils, Classes, Controls, Graphics, Windows;

type

  TLincoSwitch = class(TCustomControl)

 private

    FIsOn: Boolean;

    FPicOn: Graphics.TBitmap;

    FPicOff: Graphics.TBitmap;

    procedure FSetIsOn(AValue: Boolean);

 protected

    procedure Click;override;

    procedure Paint;override;

 public

    constructor Create(AOwner: TComponent);override;

    destructor Destroy;override;

 published

    property IsOn: Boolean read FIsOn write FSetIsOn;

    property OnClick;

    property OnKeyDown;

    property OnKeyPress;

    property OnKeyUp;

    property OnCanResize;

    property OnDblClick;

    property OnMouseDown;

    property OnMouseMove;

    property OnMouseUp;

    property OnMouseWheel;

    property OnResize;

  end;

procedure Register;

implementation

{$R *.res}

procedure LoadBitmapFromRes(ABitmapId: string; ABitmap: Graphics.TBitmap);

begin

  ABitmap.LoadFromResourceName(hInstance, ABitmapId);//从资源文件中读取位图

end;

constructor TLincoSwitch.Create(AOwner: TComponent);

begin

  inherited Create(AOwner);

  FPicOn := Graphics.TBitmap.Create;

  FPicOff := Graphics.TBitmap.Create;

  LoadBitmapFromRes('SWITCHON', FPicOn);

  LoadBitmapFromRes('SWITCHOFF', FPicOff);

  Invalidate;

end;

destructor TLincoSwitch.Destroy;

begin

  FPicOn.Free;

  FPicOff.Free;

  inherited;

end;

procedure TLincoSwitch.Click;

begin

  IsOn := not IsOn;//改变按钮的状态

  Invalidate;

  inherited;

end;

procedure TLincoSwitch.Paint;

begin

//画开关图案  

if IsOn then

    

      StretchBlt(Canvas.Handle, 0, 0, self.Width, self.Height, FPicOn.Canvas.Handle,

             0, 0, FPicOn.Width, FPicOn.Height,SRCCOPY)

  else

      StretchBlt(Canvas.Handle, 0, 0, self.Width, self.Height, FPicOff.Canvas.Handle,

             0, 0, FPicOff.Width, FPicOff.Height,SRCCOPY);

end;

procedure TLincoSwitch.FSetIsOn(AValue: Boolean);

begin

  FIson := AValue;

  Invalidate;

end;

procedure Register;

begin

  RegisterComponents('Linco', [TLincoSwitch]);

end;

end.

3、代码分析

  （1）、因为我们要在控件表面上将按钮的图案画出来，所以我们选择TcustomControl做为父类控件，因为它有个Canvas属性，我们可以利用Canvas在控件表面作图。不选用Tcontrol的原因是因为它有很多我们不需要的属性。

（2）、ABitmap.LoadFromResourceName(hInstance, ABitmapId);是从资源文件中读取Id为AbitmapId的位图，关于资源文件的使用请参考其他相关资料。注意代码中的“{$R *.res}”，它的作用是将资源文件编译到程序文件中，如果没有这个预编译条件，程序将会出现错误。

（3）、StretchBlt是将位图画到画板上，使用方法请参考MSDN。

（4）、我们为控件增加了IsOn属性。这个布尔属性用来表示开关的状态（开/关）。

从property IsOn: Boolean read FIsOn write FSetIsOn;我们可以看出这个属性是个可读可写的属性。当读这个属性时会将FisOn的值返回给调用者，而写属性时则会调用FsetIsOn方法，并将赋给属性的值做为参数传递给FsetIsOn。在FsetIsOn方法中，有如下实现代码：

  FIson := AValue;

  Invalidate;

首先将Fison设置为参数传递来的值，然后调用  Invalidate;要求重画控件，以告诉用户控件的状态已经改变，这一点是使用写字段无法做到的。

（5）    

    FPicOn: Graphics.TBitmap;

FPicOff: Graphics.TBitmap;

是声明两个.Tbitmap类型变量以保存控件的开关两种状态的图案。

（6）

    procedure Click;override;

procedure Paint;override;

分别是覆盖父类中相应的调度方法。当控件被鼠标单击时，Click方法会被调用，我们将在Click中改变控件的开关状态；Paint方法则在用户调用  Invalidate方法或控件发生重画时调用，我们一般在这个方法绘制控件的图案。

（7）、TcustomControl中又很多事件处理句柄。比如OnClick、OnKeyDown等，但是它把他们声明成了Protected保护级别，所以我们在Object Inspector中看不到他们，如果我们要他们可以在Object Inspector中被用户编辑的话，只要在Published中重新声明他们即可，不用写他们的读写方法，只要使用：Property  属性名；

这样的方法就可以。比如这个例子中的：Property Onclick;

思考题：

1、        做一个有特效的按钮控件，当鼠标按下时按钮是一个红色边框的空心圆，当鼠标松开时按钮是一个淡绿色边框的空心圆。

2、        对于父类控件中为protected的属性，如果想将它在子类控件中公布，应该怎么做？请思考Delphi为什么要将一些属性设为protected级别？

四、对特定字符串敏感的Edit控件 

我们这个控件将演示控件的自定义事件的书写。这个控件有一个类型为string的SensitiveText属性，当用户在输入框中输入的文字为InvalidText时就会触发OnSensitiveText事件。按照惯例，我先把源码展示给大家：

unit TextSenseEdit;

interface

uses

  SysUtils, Classes, Controls, StdCtrls;

type

  TSensitiveTextEvent = procedure(AText: string) of object;//方法指针

  TTextSenseEdit = class(TEdit)

  private

    FSensitiveText: string;

    FOnSensitiveText: TSensitiveTextEvent;

    procedure SetSensitiveText(AValue: string);

  protected

    procedure Change;override;

  public

  published

    property SensitiveText: string read FSensitiveText write SetSensitiveText;

    property OnSensitiveText: TSensitiveTextEvent read FOnSensitiveText write FOnSensitiveText;

  end;

procedure Register;

implementation

procedure Register;

begin

  RegisterComponents('Linco', [TTextSenseEdit]);

end;

procedure TTextSenseEdit.Change;

begin

  inherited;

  if Text = SensitiveText then

    if Assigned(OnSensitiveText) then

      OnSensitiveText(Text);

end;

procedure TTextSenseEdit.SetSensitiveText(AValue: string);

begin

  FSensitiveText := AValue;

end;

end.

代码解释：

（1）、SensitiveText属性的添加方法大家已经熟悉了，这里不多解释。

（2）、正如大家猜测的，Change方法正是编辑框文字发生变化时的调度方法，它将引起OnChange事件。我们可以在这个方法中监控编辑框文字发生的变化，当文字等于SensitiveText就触发OnSensitiveText事件（具体的实现方法在后边解释）。

（3）、Delphi中的控件的事件机制是通过方法指针来实现的。声明方法指针的格式为：

方法指针名称 = procedure(参数列表) of object;
声明事件属性的方法与声明普通属性的方法相同。在我们这个例子中，我们首先声明一个FOnSensitiveText: TSensitiveTextEvent;私有变量，然后property OnSensitiveText: TSensitiveTextEvent read FOnSensitiveText write FOnSensitiveText; 声明事件属性。这样注册控件后，当用户把控件放到窗体中后，就会在Object Inspector中Evnets页中出现OnSensitiveText事件，我们就可以像使用其他事件一样使用这个事件了。

  但是我们现在只是声明了一个事件属性，并没有书写任何代码来激发这个事件。我们应该在合适的时候激发此事件，显而易见我们应该在Change方法中激发此事件：

procedure TTextSenseEdit.Change;

begin

  inherited;

  if Text = SensitiveText then

    if Assigned(OnSensitiveText) then

      OnSensitiveText(Text);

end;

当if Text = SensitiveText时就判断控件使用者是否为OnSetSensitiveText写代码了（准确的说是是否为OnSetSensitiveText事件句柄赋值了），如果写代码了则调用OnSetSensitiveText(Text);来激发OnSetSensitiveText事件，并把控件的Text传递给方法的Avalue参数。正如“方法指针”这个名字一样，被声明为方法指针类型的变量可以当作方法使用，用来激发事件。VCL已经为我们预定义了一些常用的事件句柄，我们直接拿来使用：TnotifyEvent，TmouseEvent，TmouseMoveEvent，TkeyPressEvent等，具体可以参考VCL源码。

思考题：

做一个支持累加运算的文本编辑框控件，用户可以在编辑框中输入正整数。当用户按回车时，如果编辑框中输入的不是正整数（为负数、小数或一般字符串）则触发控件的OnError事件；如果输入的是正整数，则开始计算从1到用户输入的那个正整数中所有整数的和（用1＋2＋3+……这种累加的办法实现，不要用(1+n)*n/2这种直接计算的方法），并且在计算工程中如果发现计算的中间结果位数是5，则触发OnTailFive事件。
五、复合控件

复合控件是Delphi控件中非常重要的一种控件，复合控件就是将两个或两个以上的控件重新组合成一个新的控件。例如TspinEdit、TlabeledEdit、TDBNavigator等就是复合控件，TDBNavigator其实就是在一个Panel放上若干个Button而已。制作一个复合控件时，我们一般从TwinControl派生控件。

我们这次做的控件是拥有一个Edit编辑框和一个Button按钮的复合控件，在用户在编辑框中输入文字的过程中，Button将随时显示编辑框中文字的长度。我们把控件的源码先展示给大家。

unit EditButton;

interface

uses

  SysUtils, Classes, Controls, StdCtrls, Messages;

type

  TEditButton = class(TWinControl)

  private

    FEdit: TEdit;

    FButton: TButton;

    FText: string;

    procedure FSetText(AValue: string);

    procedure OnEditChange(Sender: TObject);

  protected

    procedure WMSize(var Msg: TMessage);message WM_SIZE;

  public

    constructor Create(AOwner: TComponent);override;

    destructor Destroy;override;

  published

    property Text: string read FText write FSetText;

  end;

procedure Register;

implementation

procedure Register;

begin

  RegisterComponents('Linco', [TEditButton]);

end;

constructor TEditButton.Create(AOwner: TComponent);

begin

  inherited;

  FEdit := TEdit.Create(nil);

  FEdit.Parent := self;

  FEdit.Top := 0;

  FEdit.Left := 0;

  FEdit.Height := Height;

  FEdit.Width := Width div 2;

  FEdit.OnChange := OnEditChange;

  FButton := TButton.Create(nil);

  FButton.Parent := self;

  FButton.Top := 0;

  FButton.Left := Width div 2;

  FButton.Height := Height;

  FButton.Width := Width div 2;

end;

destructor TEditButton.Destroy;

begin

  FEdit.Free;

  FButton.Free;

  inherited;

end;

procedure TEditButton.FSetText(AValue: string);

begin

  FEdit.Text := AValue;

end;

procedure TEditButton.OnEditChange(Sender: TObject);

begin

  FButton.Caption := IntToStr(Length(FEdit.Text));

end;

procedure TEditButton.WMSize(var Msg: TMessage);

begin

  FEdit.Height := Height;

  FEdit.Width := Width div 2;

  FButton.Left := Width div 2;

  FButton.Height := Height;

  FButton.Width := Width div 2;

end;

end.

代码解释：

（1）、我们首先定义了两个变量   

    FEdit: TEdit;

    FButton: TButton;

  分别代表复合控件中的文字编辑框和按钮。

（2）所谓复合控件说简单一点就是在一个共同的基板上将组成复合控件的各个控件（可以叫做子控件）画出来。所以我们在构造函数中建立各个子控件，然后分别设定它们的位置等属性。

以文字编辑框为例：

FEdit := TEdit.Create(nil);

的作用是建立编辑框控件。如果Create的参数指定为nil,则子控件在设计状态是可以响应用户的操作的；而如果设定为self(即设定子控件的父控件为基板)，则子控件在设计时时不可响应用户操作的，如果设定为self则析构函数中就不用Fedit.Free来销毁对象了，对象会自动销毁。
  FEdit.Parent := self;的作用是设定子控件的父控件，如果没有这一句则控件是无法显示的。

  FEdit.Top := 0;

  FEdit.Left := 0;

  FEdit.Height := Height;

  FEdit.Width := Width div 2;

这四句是设定控件在基板上的相对位置的，这里的Top,Left不是相对于窗体的，而是相对于基板的。
  FEdit.OnChange := OnEditChange;

则是设定编辑框控件的OnChange(文字改变事件)的处理句柄为OnEditChange；

（1）    用户有可能在设计时或运行时通过代码改变控件的大小，这时控件中子控件的顺序就会变得乱七八糟，所以需要相应控件的WM_SIZE事件（控件大小发生变化的事件）重新设定子控件的位置，大小等。函数WMSize的作用就是这样的。

安装控件后发现控件已经可以正确运行了，但是还有一个问题，就是这个控件没有了Onclick,Onchange等必须的属性。我们只要为控件增加事件处理句柄属性，然后把事件处理句柄属性的读写方法都指向子控件的事件处理句柄属性即可。例如我们为控件增加OnClick事件，这个事件发生在用户单击按钮时，我么只要在Pulished部分增加如下代码：

property  OnClick:  TnotifyEvent read GetOnClick write SetOnClick

在Private中增加如下方法声明：

function GetOnclick: TnotifyEvent;

procedure SetOnclick(AValue: TnotifyEvent);
这两个方法的实现分别为：

function  TeditButton. GetOnclick: TnotifyEvent;

begin

  result := Fbutton.Onclick;

end;

procedure TeditButton. SetOnclick(AValue: TnotifyEvent); 

begin

  Fbutton.OnClick := Avalue;

end;
思考题：

1、做一个模仿播放器中的操作按钮的复合控件，控件由三个按钮组成，分别是“播放”、“暂停”、“停止”，请按照正常的逻辑关系，处理这三个按钮的可用/不可用关系。（提示：可以参考TDBNavigator的源代码）

六、控件手拉手――控件关联的实现

控件的关联在Delphi中也是很常见的，我们可以设定一个控件的某个属性指向另一个控件。比如我们在窗体上放上Tedit,TpopupMenu两个控件，然后设定Tedit的PopupMenu属性为TpopupMenu控件，运行后在Tedit点击右键就会弹出刚才设定的那个TpopupMenu菜单，也就是说Tedit,TpopupMenu联手完成了任务。再比如TDBEdit控件的DataSource属性就可以指向一个TdataSource控件，这样就可以在TDBEdit控件中显示TdataSource输出的某个字段的值了。

  下面我们将写一个简单的实现控件关联的控件。这个控件派生于Tedit,它可以与一个Tlabel控件关联，在控件的编辑框中输入文字时，与它关联的Tlabel控件的文字将随着它而变化。代码如下：

unit MyEdit;

interface

uses

  SysUtils, Classes, Controls, StdCtrls;

type

  TMyEdit = class(TEdit)

  private

    FLinkLabel: TLabel;

    procedure FSetLinkLabel(AValue: TLabel);

  protected

    procedure Notification(AComponent: TComponent;Operation: TOperation);

          override;

    procedure Change;override;

  public

  published

    property LinkLabel: TLabel read FLinkLabel write FSetLinkLabel;

  end;

procedure Register;

implementation

procedure Register;

begin

  RegisterComponents('Linco', [TMyEdit]);

end;

procedure TMyEdit.Change;

begin

  inherited;

  if LinkLabel <> nil then

    LinkLabel.Caption := Text;

end;

procedure TMyEdit.FSetLinkLabel(AValue: TLabel);

begin

  FLinkLabel := AValue;

  if AValue <> nil then

    FLinkLabel.FreeNotification(self);

end;

procedure TMyEdit.Notification(AComponent: TComponent;

  Operation: TOperation);

begin

  inherited;

  if (Operation = opRemove) and (AComponent = LinkLabel) then

    LinkLabel := nil;

end;

end.

代码解释：

（1）、我们只要将控件的任意一个属性的类型设定为另外一个控件的类名称，那么我们就可以在控件的Object Inspector中将这个属性指向那个控件（或那个控件的派生控件）的一个实例。比如本例中我们增加了LinkLabel属性，它的类型为 Tlabel，所以我们就可以把LinkLabel属性指向一个标签控件。

（2）、请注意FsetLinkLabel中的这段代码： 

if AValue <> nil then

FLinkLabel.FreeNotification(self);

如果我们将控件关联属性指向了一个控件，可是后来又将被指向的控件删除了，那么我们的控件关联属性是不会自动删除的，这样就会造成控件关联属性指向的控件不存在的现象。我们必须自动感知被关联控件的删除并重新设定控件关联属性为不指向任何控件，这样就避免了错误的发生。

FLinkLabel.FreeNotification(self);的作用就是这样的。它调用控件的FreeNotification方法（在Tcomponent中定义）向被指向的控件注册一个“消息”，当被指向控件被删除时，会向所有向他注册的控件发送一个它被删除的消息，此时向他注册的控件就会触发Notification方法，这样我们就可以自动感知被指向控件的状态了。这是设计模式中Observer(观察者)模式的典型应用。

既然向他注册的控件就会触发Notification方法，我们就覆盖父类的Notification方法，写出如下的代码：

  if (Operation = opRemove) and (AComponent = LinkLabel) then 
LinkLabel := nil;

这句话的意思是：如果控件被删除并且被删除的控件（因为我们的控件可能向多个控件注册了消息）是LinkLabel，那么我们就设定LinkLabel属性不指向任何控件。

（3）覆盖父类的Change调度方法。在此方法里为连接的LinkLabel的Caption赋值就达到我们的目的了。
思考题：

1、做一个Label控件，给它增加一个DataSource属性，该属性可以指向一个TdataSource类型的控件，它有一个GetRecordCount方法。当调用此方法时，就在Label控件中显示这个DataSource对应的数据集中的记录的条数。
对话框控件的制作

Delphi中有很多对话框组件，例如TopenDialog、TfontDialog等。这些控件的特点就是虽然是不可视控件，但是在运行时都有一个可视化的效果，比如TopenDialog的可视化效果就是一个打开对话框。我们这次将开发一个日期对话框控件，当我们调用控件的Execute方法（不一定非要使用Execute方法，不过大部分对话框控件都是使用这个方法，我们也就按照惯例来了）时，就会弹出一个可以选择日期的对话框，我们选择一个日期后，点击“确定”则Execute返回True,点击“取消”则Execute返回False。我们可以读取Date属性来得到用户选择的日期，也可以修改此属性来改变对话框的初始日期。
1、新建一个对话框。在对话框窗体上放置一个TmonthCalendar组件，命名为Cal,窗体名称改为FormDate。在窗体上放置两个按钮，一个按钮的Caption为“确定(&O)”，ModalResult为mrOk，一个按钮的Caption为“取消(&C)”，ModalResult为mrCancel。设计好的窗体如下图所示：

[image: image3.jpg]



2、为窗体添加两个Public访问级的方法：

    function GetSelDate: TDate;

    procedure SetInitDate(AValue: TDate);

代码如下：

function TFormDate.GetSelDate: TDate;

begin

  result := cal.Date;

end;

procedure TFormDate.SetInitDate(AValue: TDate);

begin

  cal.Date := AValue;

end;
3、新建一个控件，派生自Tcomponent。

代码如下：

unit DateDialog;

interface

uses

  SysUtils, Classes, Controls, frmDlg;

type

  TDateDialog = class(TComponent)

  private

    FDlg: TFormDate;

    function GetDate: TDate;

    procedure SetDate(AValue: TDate);

  protected

  public

    constructor Create(AOwner: TComponent);override;

    function Execute: Boolean;

  published

    property Date: TDate read GetDate write SetDate;

  end;

procedure Register;

implementation

procedure Register;

begin

  RegisterComponents('Linco', [TDateDialog]);

end;

constructor TDateDialog.Create(AOwner: TComponent);

begin

  inherited Create(AOwner);

  FDlg := TFormDate.Create(self);

end;

function TDateDialog.Execute: Boolean;

begin

  result := (FDlg.ShowModal = mrOK);

end;

function TDateDialog.GetDate: TDate;

begin

  result := FDlg.GetSelDate;

end;

procedure TDateDialog.SetDate(AValue: TDate);
begin

  FDlg.SetInitDate(AValue);

end;

end.

代码比较简单就不多解释了。
思考题：

1、做一个模仿TcolorDialog的对话框控件。

